Code 83-24374-Opt2: 

“Apply a Route Layer's Definition Query to a Route Event layer”-ArcGIS 8.3 only
Public Sub AddPointEventLayer()

  '+++ This code shows how to add a point event layer to the map document. If defined, the route

  '+++ layer's definition query will be copied and applied to the route locator. This will limit

  '+++ the number of routes events will be displayed on.

  '+++ You should modify the code to suit your data.

  On Error GoTo eh

  '+++ Get the event table. It is called 'accident' in ArcMap's table of contents.

  Dim pMxDoc As IMxDocument

  Dim pTblColl As ITableCollection

  Dim pEventTable As ITable

  Dim i As Long

  Dim pMap As IMap

  Dim pDS As IDataset

  Set pMxDoc = ThisDocument

  Set pMap = pMxDoc.FocusMap

  Set pTblColl = pMap

  For i = 0 To pTblColl.TableCount - 1

    Set pDS = pTblColl.Table(i)

    If LCase(pDS.BrowseName) = "accident" Then '+++ change this line to match your data

      Set pEventTable = pDS

      Exit For

    End If

  Next i

  If pEventTable Is Nothing Then

    MsgBox "Could not find event table", vbExclamation, "AddPointEventLayer"

    Exit Sub

  End If

'+++ Get the route feature layer. It is called 'routes' in ArcMap's table of contents.

  Dim pLayer As ILayer

  Dim pFLayer As IFeatureLayer

  For i = 0 To pMap.LayerCount - 1

    Set pLayer = pMap.Layer(i)

    If LCase(pLayer.Name) = "routes" Then '+++ change this line to match your data

        If TypeOf pLayer Is IFeatureLayer Then

          Set pFLayer = pLayer

          Exit For

        End If

    End If

  Next i

  If pFLayer Is Nothing Then

    MsgBox "Could not find the route layer", vbExclamation, "AddPointEventLayer"

    Exit Sub

  End If

  '+++ Create a route event source object. This will act as the basis of the new event layer ...

  '+++ 1. Get the route layer's route locator object. It will respect the route layer's definition query

  Dim pLayerExt As ILayerExtensions

  Dim pRtLayerExt As IRouteLayerExtension

  Dim pRtLocName As IRouteLocatorName

  Set pLayerExt = pFLayer

  For i = 0 To pLayerExt.ExtensionCount - 1

    If TypeOf pLayerExt.Extension(i) Is IRouteLayerExtension Then

      Set pRtLayerExt = pLayerExt.Extension(i)

      Exit For

    End If

  Next i

  If pRtLayerExt Is Nothing Then

    MsgBox "The route layer does not have a route locator attached to it." & vbCrLf & _

       "Perhaps this layer was created before ArcGIS 8.3.", vbExclamation

    Exit Sub

  Else

Set pRtLocName = pRtLayerExt.RouteLocatorName(pFLayer)

  End If

  '+++ 2. Create the route event table properties.

  Dim pRtProp As IRouteEventProperties2

  Dim pRMPtProp As IRouteMeasurePointProperties2

  Set pRtProp = New RouteMeasurePointProperties

  With pRtProp

    .EventMeasureUnit = esriUnknownUnits

    .EventRouteIDFieldName = "rkey" '+++ change this line to match your data

    '.LateralOffsetFieldName = "lat_offset"

    .AddErrorField = True             '+++ adds field for locating errors

    .ErrorFieldName = "LOC_ERROR"     '+++ the name for the locating errors field

  End With

  Set pRMPtProp = pRtProp

  With pRMPtProp

    .MeasureFieldName = "mile"  '+++ change this line to match your data

    .AddAngleField = True

    .AngleFieldName = "LOC_ANGLE"

    .AsPointFeature = True        '+++ Point events will be points. Multipoint if False

    .NormalAngle = True           '+++ The angle normal to the digitize direction.  Tangent if False

    .ComplementAngle = False

  End With

'+++ 3. Create the route event source name object

  Dim pName As IName

  Set pDS = pEventTable

  Set pName = pDS.FullName

  Dim pRESN As IRouteEventSourceName

  Set pRESN = New RouteEventSourceName

  With pRESN

    Set .EventTableName = pName

    Set .EventProperties = pRMPtProp

    Set .RouteLocatorName = pRtLocName

  End With

  '+++ Create a new feaure layer and set it's feature class to be the route event source

  Set pName = pRESN

  Set pFLayer = New FeatureLayer

  Set pFLayer.FeatureClass = pName.Open

  pFLayer.Name = pDS.BrowseName + "_Events"  '+++ "Accident_Events"

  pMap.AddLayer pFLayer

  pMxDoc.UpdateContents

  pMxDoc.ActiveView.Refresh

  Exit Sub

eh:

  Dim lNum As Long, sDesc As String, sSrc As String

  lNum = Err.Number

  sDesc = Err.Description

  sSrc = Err.Source

  Err.Raise lNum, sSrc, sDesc

End Sub

